Examples of Creative Telescoping

```
> with(CreativeTelescoping):
```

Legendre polynomials

```
> F:=Sum(2^(-n)*binomial(n,k)*binomial(n,n-k)*(x+1)^k*(x-1)^(n-k),
  k=0..n);
```

$$F := \sum_{k=0}^{n} 2^{-n} \binom{n}{k} \binom{n}{n-k} (x+1)^k (x-1)^{n-k} \tag{1.1}$$

```
> CreativeTelescoping(F,[n::shift,x::diff],certificate='cert');
```

$$\left[D_n (n+1) + (-x^2+1) D_x - n x - x, D_x^2 (x^2-1) + 2 D_x x - n^2 - n\right] \tag{1.2}$$

```
> normal(cert);
```

$$\left[\frac{(x-1) k^2 (2k - 3n - 3)}{2 (k^2 - 2nk + n^2 - 2k + 2n + 1)}, \frac{2k^2}{x+1}\right] \tag{1.3}$$

Chebyshev coefficients of exp(-px)

```
> F:=Int(exp(-p*x)*ChebyshevT(n,x)/sqrt(1-x^2),x=-1..1);
```

$$F := \int_{-1}^{1} \frac{e^{-px} \operatorname{ChebyshevT}(n,x)}{\sqrt{-x^2+1}} \, dx \tag{2.1}$$

```
> CreativeTelescoping(F,[p::diff,n::shift]);
```

$$\left[p D_n + D_p p - n, p D_n^2 - 2 D_n n - p - 2 D_n\right] \tag{2.2}$$

A bivariate mixed integral

```
> F:=Int((1+x/(n^2+1))*((x+1)^2/(x-4)/(x-3)^2/(x^2-5)^3)^n*sqrt
  (x^2-5)*exp((x^3+1)/x/(x-3)/(x-4)^2),x);
```

$$F := \int \left(1 + \frac{x}{n^2+1}\right) \left(\frac{(x+1)^2}{(x-4)(x-3)^2(x^2-5)^3}\right)^n \sqrt{x^2-5}\, e^{\frac{x^3+1}{x(x-3)(x-4)^2}} \, dx \tag{3.1}$$

```
> CreativeTelescoping(F,[n::shift]);
```

$$\left[95004[\text{...171 digits...}]00000\, n^{89} D_n^8 + 10312[\text{...167 digits...}]00000\, n^{90} D_n^6 \right. \tag{3.2}$$

$+ 94178[...172\ digits...]00000\, n^{89}\, D_n^7 + 80684[...178\ digits...]00000\, n^{88}\, D_n^8$
$+ 82759[...173\ digits...]00000\, n^{87}\, D_n^9 + 10155[...168\ digits...]00000\, n^{90}\, D_n^5$
$+ 90151[...173\ digits...]00000\, n^{89}\, D_n^6 + 79982[...179\ digits...]00000\, n^{88}\, D_n^7$
$+ 88363[...181\ digits...]00000\, n^{87}\, D_n^8 + 70284[...180\ digits...]00000\, n^{86}\, D_n^9$
$+ [...884\ terms...] + 13823[...219\ digits...]00000\, D_n^4 + 32493[...205\ digits...]00000\, n^3$
$+ 22617[...209\ digits...]00000\, n^2\, D_n + 15355[...213\ digits...]00000\, n\, D_n^2$
$- 64898[...215\ digits...]00000\, D_n^3 + 28507[...204\ digits...]00000\, n^2$
$- 52055[...207\ digits...]00000\, n\, D_n + 66053[...211\ digits...]00000\, D_n^2$
$+ 10134[...203\ digits...]00000\, n - 98869[...206\ digits...]00000\, D_n\Big]$

```
> degree(op(%),D[n]);
```

$$9 \tag{3.3}$$

```
> collect(op(%%),D[n],degree);
```

$$87\, D_n^9 + 89\, D_n^8 + 89\, D_n^7 + 90\, D_n^6 + 90\, D_n^5 + 90\, D_n^4 + 90\, D_n^3 + 90\, D_n^2 + 90\, D_n + 90 \tag{3.4}$$

Generating function of the Jacobi polynomials

```
> S := Sum(JacobiP(n, a,b,x)*z^n/n!, n = 0 .. infinity)
```

$$S := \sum_{n=0}^{\infty} \frac{\mathrm{JacobiP}(n, a, b, x)\, z^n}{n!} \tag{4.1}$$

```
> T:=CreativeTelescoping(S, [x::diff, a::shift,b::shift, z::diff],
  certificate='cert');
```

$$T := \Big[2 + (x-1)\, D_a + (-x-1)\, D_b,\ (a\,x + x\,b + a + b)\, D_b + \left(-2\,x^2 + 2\right) D_x + 2\, D_z^2\, z + (-2\,x\,z + 2)\, D_z - 2\,a\,x - 2\,x\,b - 2\,a - 2\,x,\ z\, D_z\, D_b + (a + b + 1)\, D_b + (1 - x)\, D_x - D_z\, z - a - b - 1,\ (a\,x + x\,b + a - b)\, D_b + (2\,x - 2)\, D_z\, D_x + \left(-2\,x^2 + 2\,x\right) D_x + (-2\,x\,z + 2\,z)\, D_z - 2\,a\,x - 2\,x\,b + 2\,b - 2\,x + 2,\ (x\,z + z)\, D_b^2 + (2\,a + 2\,b - 2\,z + 2)\, D_b + (-2\,x + 2)\, D_x - 2\,a - 2\,b - 2,\ \left(x^2 - 1\right) D_b\, D_x + (a\,x + x\,b + a - b + x - 1)\, D_b + \left(-x^2 + 1\right) D_x + (-x\,z + z)\, D_z - a\,x - x\,b - a + b - x + 1,\ (a\,x\,z + b\,x\,z + a\,z + b\,z)\, D_b + \left(2\,x^2 - 2\right) D_x^2 + \left(-2\,z\,x^2 + 2\,a\,x + 2\,x\,b + 2\,a - 2\,b + 4\,x + 2\,z\right) D_x + \left(-2\,x\,z^2 - 2\,a\,z - 2\,b\,z - 2\,z\right) D_z - 2\,a\,x\,z - 2\,b\,x\,z - 2\,a\,z - 2\,x\,z\Big] \tag{4.2}$$

```
> map(EvalCert,cert,S,[x::diff, a::shift,b::shift, z::diff],[n=0])
  ;
```

$$[0, 0, 0, 0, 0, 0, 0] \tag{4.3}$$

```
> map(SingularitiesCertificate,cert,n);
```

$$\big[(a + b + 2\,n + 2)\,(a + b + 2\,n)\,(a + b - 2 + 2\,n),\ (a + b + 2\,n + 2)\,(a + b + 2\,n)\,(a + b - 2 + 2\,n),\ (a + b + 2\,n + 2)\,(a + b + 2\,n)\,(a + b - 2 + 2\,n),\ (a \tag{4.4}$$

$+ b + 2n + 2)(a + b + 2n)(a + b - 2 + 2n), (2n + 3 + a + b)(a + b + 1$
$+ 2n)(a + b - 1 + 2n)(a + b + 2n + 2)(a + b + 2n)(a + b - 2 + 2n), (a$
$+ b + 2n + 2)(a + b + 2n)(a + b - 2 + 2n), (a + b + 2n + 2)(a + b$
$+ 2n)(a + b - 2 + 2n)]$

Sum of BesselJ(k,x)^2

```
> S := Sum(BesselJ(n, x)^2, n = 1 .. infinity)
```

$$S := \sum_{n=1}^{\infty} \mathrm{BesselJ}(n, x)^2 \quad (5.1)$$

```
> T := CreativeTelescoping(S, [x::diff], certificate = 'cert');
```

$$T := [D_x] \quad (5.2)$$

```
> normal(cert);
```

$$\left[\frac{-x^2 D_n^2 + 4n^2 D_n - 8n^2 + 8n D_n + x^2 - 8n + 4 D_n}{4x(n+1)}\right] \quad (5.3)$$

```
> EvalCert(cert[1],S,[x::diff]);
```

$$-\frac{\mathrm{BesselJ}(n, x)\,(2n\,\mathrm{BesselJ}(n, x) - \mathrm{BesselJ}(n+1, x)\,x)}{x} \quad (5.4)$$

```
> simplify(eval(%,n=1),BesselJ);
```

$$-\mathrm{BesselJ}(1, x)\,\mathrm{BesselJ}(0, x) \quad (5.5)$$

```
> diff(BesselJ(0,x)^2,x);
```

$$-2\,\mathrm{BesselJ}(1, x)\,\mathrm{BesselJ}(0, x) \quad (5.6)$$

Conclusion:

```
> Diff(1/2*BesselJ(0,x)^2+S,x)=0
```

$$\frac{\partial}{\partial x}\left(\frac{\mathrm{BesselJ}(0, x)^2}{2} + \left(\sum_{n=1}^{\infty} \mathrm{BesselJ}(n, x)^2\right)\right) = 0 \quad (5.7)$$

It has to be a constant, that can be found from the initial conditions:

```
> eval(BesselJ(0,x)^2/2,x=0);
```

$$\frac{1}{2} \quad (5.8)$$

Thus, we recover the classical identity

```
> BesselJ(0,x)^2+2*S=1;
```

$$\mathrm{BesselJ}(0, x)^2 + 2\left(\sum_{n=1}^{\infty} \mathrm{BesselJ}(n, x)^2\right) = 1 \quad (5.9)$$

Check:

```
> series(BesselJ(0,x)^2+2*add(BesselJ(i,x)^2,i=1..10),x,10);
```

$$1 + O(x^{10}) \quad (5.10)$$

Apéry's sum

```
> Sap := Sum(binomial(n, k)^2*binomial(n + k, k)^2, k = 0 .. n)
```

$$Sap := \sum_{k=0}^{n} \binom{n}{k}^2 \binom{n+k}{k}^2 \quad \textbf{(6.1)}$$

```
> T:=CreativeTelescoping(Sap, [n::shift], certificate = 'cert');
```

$$T := \left[\left(n^3 + 6 n^2 + 12 n + 8\right) D_n^2 + \left(-34 n^3 - 153 n^2 - 231 n - 117\right) D_n + n^3 + 3 n^2 + 3 n + 1 \right] \quad \textbf{(6.2)}$$

When we sum both sides of the telescoping equation, for k from 0 to n+2 (so as to obtain S_{n+2}), we need to evaluate the certificate at the extremities 0 and n+3:

```
> EvalCert(cert[1],Sap,[n::shift],[k=0]);
```

$$0 \quad \textbf{(6.3)}$$

```
> EvalCert(cert[1],Sap,[n::shift],[k=n+3]);
```

$$-\left(4 n^7 + 60 n^6 + 367 n^5 + 1161 n^4 + 1962 n^3 + 1566 n^2 + 243 n - 243\right) \binom{n}{n+3}^2 \binom{2 n+3}{n+3}^2 \quad \textbf{(6.4)}$$

```
> simplify(%) assuming n::posint;
```

$$0 \quad \textbf{(6.5)}$$

The summation of the certificate at the intermediate values in the range must also be possible. But:

```
> factor(normal(cert));
```

$$\left[\frac{1}{(-n-1+k)^2 (k-n-2)^2} \left(4 k^4 \left(3 k^3 D_n - 10 k^2 n D_n + 11 k n^2 D_n - 4 n^3 D_n - 3 k^3 + 2 k^2 n - 14 k^2 D_n + 5 k n^2 + 30 k n D_n - 4 n^3 - 16 n^2 D_n + 8 k^2 + 12 n k + 19 k D_n - 20 n^2 - 20 n D_n + 4 k - 32 n - 8 D_n - 16\right)\right) \right] \quad \textbf{(6.6)}$$

it has poles at k=n+1 and k=n+2. In this case one can incorporate these in the binomial coefficients. In general, having the certificate allows to check by series expansion that the summand is actually finite at these points:

```
> Order:=3:
> map(normal,EvalCert(cert[1],Sap,[n::shift],[k=n+1],series=true))
  ;
```

$$-\frac{4 \left(n^2 + 2 n + 1\right) \left(4 n^5 + 36 n^4 + 111 n^3 + 157 n^2 + 105 n + 27\right) \Gamma(2 n+2)^2}{\Gamma(n+2)^4} + O\left(\left(-n-1+k\right)\right) \quad \textbf{(6.7)}$$

```
> map(normal,EvalCert(cert[1],Sap,[n::shift],[k=n+2],series=true))
  ;
```

$$-\frac{4 (n+2)^3 \left(n^2 + 4 n + 4\right) \left(4 n^2 + 12 n + 9\right) \Gamma(2 n+3)^2}{\Gamma(n+3)^4} + O\left(\left(k-n-2\right)\right) \quad \textbf{(6.8)}$$

and thus the telescoper st[1] cancels the sum.

Another way

Another, maybe simpler, way is to compute the summation up to k=n-1: Sum(telesc(u_k),k=0..n-1)= cert(u_n), then add the missing summands to get telesc(S).

```
> U:=op(1,Sap);
```

$$U := \binom{n}{k}^2 \binom{n+k}{k}^2 \quad \textbf{(6.1.1)}$$

```
> add(coeff(T[1],D[n],i)*subs(n=n+i,U),i=0..2);
```

$$\left(n^3+3n^2+3n+1\right)\binom{n}{k}^2\binom{n+k}{k}^2+\left(-34n^3-153n^2-231n-117\right)\binom{n+1}{k}^2\binom{n+k+1}{k}^2+\left(n^3+6n^2+12n+8\right)\binom{n+2}{k}^2\binom{n+k+2}{k}^2 \tag{6.1.2}$$

If one sums for k from 0 to n-1, one gets cert(u_n) on the right-hand side. On the left-hand side, this is telesc(S) up to

```
> coeff(T[1],D[n],0)*eval(U,k=n)+coeff(T[1],D[n],1)*add(eval(subs
  (n=n+1,U),k=n+i),i=0..1)+coeff(T[1],D[n],2)*add(eval(subs(n=n+2,
  U),k=n+i),i=0..2);
```

$$\left(n^3+3n^2+3n+1\right)\binom{2n}{n}^2+\left(-34n^3-153n^2-231n-117\right)\left((n+1)^2\binom{1+2n}{n}^2+\binom{2n+2}{n+1}^2\right)+\left(n^3+6n^2+12n+8\right)\left(\binom{n+2}{n}^2\binom{2n+2}{n}^2+(n+2)^2\binom{2n+3}{n+1}^2+\binom{2n+4}{n+2}^2\right) \tag{6.1.3}$$

```
> normal(expand(%));
```

$$n^4\binom{2n}{n}^2\left(4n^3+36n^2+61n+24\right) \tag{6.1.4}$$

which is exactly compensated by cert(u_n):

```
> EvalCert(cert[1],Sap,[n::shift],[k=n]);
```

$$-n^4\binom{2n}{n}^2\left(4n^3+36n^2+61n+24\right) \tag{6.1.5}$$

Strehl's identity

```
> S := Sum(binomial(n+k,k)*binomial(n,k)*binomial(k,j)^3, j = 0 ..
  k);
```

$$S:=\sum_{j=0}^{k}\binom{n+k}{k}\binom{n}{k}\binom{k}{j}^3 \tag{7.1}$$

The option LFSolbasis in this example stores into the variable (here lfs) a linear functional system corresponding to the certificate. It can then be used for iterated summation.

```
> st:= CreativeTelescoping(S, [n::shift,k::shift],certificate=
  'cert', LFSolbasis='lfs');
```

$$st:=\left[D_n(-n-1+k)+n+k+1,\ \left(k^4+8k^3+24k^2+32k+16\right)D_k^2+\left(7k^4-7k^2n^2+42k^3-7k^2n-21kn^2+93k^2-21nk-16n^2+90k-16n+32\right)D_k-8k^4+16k^2n^2-8n^4-32k^3+16k^2n+32kn^2-16n^3-40k^2+32nk+8n^2-16k+16n\right] \tag{7.2}$$

So the sum satisfies

```
> lfs;
```

$$LFSol\Big(\Big\{(n+k+1)\,\_f(n,k)+(-n-1+k)\,\_f(n+1,k),\ \left(-8k^4+16k^2n^2-8n^4-32k^3+16k^2n+32kn^2-16n^3-40k^2+32nk+8n^2-16k+16n\right)\_f(n,k) \tag{7.3}$$

$+ (7 k^4 - 7 k^2 n^2 + 42 k^3 - 7 k^2 n - 21 k n^2 + 93 k^2 - 21 n k - 16 n^2 + 90 k - 16 n + 32)$ _f$(n, k+1) + (k^4 + 8 k^3 + 24 k^2 + 32 k + 16)$ _f$(n, k+2)\})$

It can now be summed over k

```
> S2:=Sum(lfs,k=0..n);
```

$$S2 := \sum_{k=0}^{n} LFSol(\{(n+k+1)\,\_f(n,k) + (-n-1+k)\,\_f(n+1,k), (-8 k^4 + 16 k^2 n^2 - 8 n^4 - 32 k^3 + 16 k^2 n + 32 k n^2 - 16 n^3 - 40 k^2 + 32 n k + 8 n^2 - 16 k + 16 n)\,\_f(n,k) + (7 k^4 - 7 k^2 n^2 + 42 k^3 - 7 k^2 n - 21 k n^2 + 93 k^2 - 21 n k - 16 n^2 + 90 k - 16 n + 32)\,\_f(n,k+1) + (k^4 + 8 k^3 + 24 k^2 + 32 k + 16)\,\_f(n,k+2)\}) \quad \textbf{(7.4)}$$

```
> st := CreativeTelescoping(S2, [n::shift],certificate='cert')
```

$$st := [(n^3 + 6 n^2 + 12 n + 8) D_n^2 + (-34 n^3 - 153 n^2 - 231 n - 117) D_n + n^3 + 3 n^2 + 3 n + 1] \quad \textbf{(7.5)}$$

This is Apéry's recurrence again, and the identity

```
> Sum(S,k=0..n)=Sap
```

$$\sum_{k=0}^{n} \sum_{j=0}^{k} \binom{n+k}{k} \binom{n}{k} \binom{k}{j}^3 = \sum_{k=0}^{n} \binom{n}{k}^2 \binom{n+k}{k}^2 \quad \textbf{(7.6)}$$

can be checked by verifying two initial conditions.

```
> eval(%,[n=0,Sum=add]),eval(%,[n=1,Sum=add]);
```

$$1 = 1, 5 = 5 \quad \textbf{(7.7)}$$